

- 1. IDENTIFICATION
- 1.1 Digital-8-21-U-Sym
- 1.2 Symbolic Tape Format Generator
- 1.3 May 2, 1965

2. ABSTRACT

The format generator allows the user to create PDP-8 symbolic tapes with formatting. It may be used to condense tapes with spaces by inserting tabs, or merely to align tabs, instructions, and comments.

3. REQUIREMENTS

3.1 Storage

Uses 4096-word core memory for buffering and program storage. The program itself occupies locations 0020-0665.

3.3 Equipment

PDP-8 with ASR 33 or with 750/75A High-Speed Reader/Punch.

4. USAGE

4.1 Loading

The program is loaded with the Binary Loader (see Digital-8-2-U).

4.3 Switch Settings

There are two switch options:

- 4.3.1 Bit 0 : If bit 0 is up, input and output are by means of the high speed reader and punch.

If bit 0 is down, input and output are by means of the ASR 33.

- 4.3.2 Bit 11: If bit 11 is up, tabs, followed by rubouts, are generated. If bit 11 is down, spaces are generated.

4.4 Start up/Entry

- 4.4.1 Load binary tape.

- 4.4.2 Set C(SR) = 0200 (starting address of program).

- 4.4.3 Press LOAD ADDRESS.

- 4.4.4 Set C(SR) for options (see 4.3).

- 4.4.5 Turn on punch; put symbolic tape in reader; turn on reader.

- 4.4.6 Press START.

- 4.4.7 When using high-speed I/O, the program will halt at the end of the tape. Using ASR 33, program will not halt.

5. RESTRICTIONS (Not Applicable)

6. DESCRIPTION

6.1 The format generator uses a fixed algorithm that is explained in the flow diagram.

6.2 The following is an example of its usage. It is the original input symbolic:

```
*200
/EXAMPLE OF FORMAT
/GENERATOR
BEGIN,0/START OF PROGRAM
KCC
KSF/WAIT FOR FLAG
JMP .-1/FLAG NOT SET YET
KRB/READ IN CHARACTER
DCA CHAR
TAD CHAR
TAD MSPACE/IS IT A SPACE?
SNA CLA
HLT/YES
JMP BEGIN+2/NO INPUT AGAIN
CHAR,0/TEMPORARY STORAGE
MSPACE,-240/-ASCII EQUIVALENT

/END OF EXAMPLE
$
```

This is the output symbolic:

| | |
|--------------------|---------------------|
| *200 | |
| /EXAMPLE OF FORMAT | |
| /GENERATOR | |
| BEGIN, | 0 /START OF PROGRAM |
| KCC | |
| KSF | /WAIT FOR FLAG |
| JMP .-1 | /FLAG NOT SET YET |
| KRB | /READ IN CHARACTER |
| DCA CHAR | |
| TAD CHAR | |
| TAD MSPACE | /IS IT A SPACE? |
| SNA CLA | |
| HLT | /YES |

```
JMP BEGIN+2      /NO INPUT AGAIN
CHAR,            0 /TEMPORY STORAGE
MSPACE,         - 240 /-ASCII EQUIVALENT
/END OF EXAMPLE
$
```

7. METHODS (Not Applicable)

8. FORMAT

8.1 Input Data

The program will accept blank tape, or code 200 for Leader/Trailer. Rubouts are removed except when preceded by tabs. Any legal symbolic tape for the PAL II or MACRO assemblers will be correctly formatted.

8.3 Output Data

Output is compatible with either assembler and with the Symbolic Tape Editor.

9. EXECUTION TIME

Limited by input/output time.

10. PROGRAM

10.4 PALII Assembly Listing

```

/PAGE 1
*20
/CONSTANTS
0020 1000 RBGN, 1000
0021 0400 HIRDPT, HIRD
0022 7540 MSPACE, -240
0023 0000 FIELD1, 0
0024 0000 CHAR, 0
0025 0000 FCOUNT, 0
0026 0377 C377, 0377
0027 0211 TAB, 0211
0030 7766 MTAB1, -12
0031 7775 M3, -3
0032 7564 MFFFEED, -214
0033 0014 MLEADR, 214-200
0034 7521 MSLASH, -257
0035 7563 MCR, -215
0036 0215 CARRTN, 215
0037 0212 LFEED, 212
0040 0000 RCNT, 0
0041 7526 MASTER, -252
0042 1400 RKON, 1000-7400
0043 7566 MLFEED, -212
0044 7401 MRBOUT, -377
0045 7567 MTAB, -211
0046 0240 SPACE, 240
0047 0000 TEM6, 0
0050 0000 SWT1, 0
0051 0000 COMMFL, 0
0052 0000 EQFLAG, 0
0053 7524 MCOMMA, -254
0054 0000 GET, 0
0055 7200 CLA
0056 1413 TAD I 13
0057 3024 DCA CHAR
0060 5454 JMP I GET
0061 0000 SPIGNR, 0
0062 4054 JMS GET
0063 1024 TAD CHAR
0064 1022 TAD MSPACE
0065 7650 SNA CLA
0066 5062 JMP SPIGNR+1
0067 5461 JMP I SPIGNR
0070 0600 RDLINE, READL

/GET A CHARACTER FROM
/LINE BUFFER

/ROUTINE TO IGNORE SPACES

```

| | | | | |
|------|------|---------|--------------|-------------------------|
| | | /PAGE 2 | | |
| | | *200 | | |
| 0200 | 7240 | | CLA CMA | /SET UP INPUT ROUTINE |
| 0201 | 3040 | | DCA RCNT | |
| 0202 | 3050 | | DCA SWTI | |
| 0203 | 1030 | START, | TAD MTAB1 | /SET UP TAB COUNT |
| 0204 | 3025 | | DCA FCOUNT | |
| 0205 | 3023 | | DCA FIELD1 | |
| 0206 | 4470 | | JMS I RDLINE | /READ IN A LINE |
| 0207 | 1051 | | TAD COMMFL | /COMMA SEEN?? |
| 0210 | 7640 | | SZA CLA | |
| 0211 | 5233 | | JMP PGO | /YES |
| 0212 | 1052 | | TAD EQFLAG | /EQUAL SEEN?? |
| 0213 | 7640 | | SZA CLA | |
| 0214 | 5310 | | JMP ECHO-1 | /YES ECHO LINE |
| 0215 | 4061 | | JMS SPIGNR | |
| 0216 | 1024 | | TAD CHAR | /CHARACTER ="*"? |
| 0217 | 1041 | | TAD MASTER | |
| 0220 | 7650 | | SNA CLA | |
| 0221 | 5311 | | JMP ECHO | /YES ECHO |
| 0222 | 1024 | | TAD CHAR | |
| 0223 | 1034 | | TAD MSLASH | |
| 0224 | 7650 | | SNA CLA | /CHARACTER ="/"? |
| 0225 | 5311 | | JMP ECHO | /YES-ECHO |
| 0226 | 1024 | | TAD CHAR | |
| 0227 | 1035 | | TAD MCR | /CHARACTER =CR? |
| 0230 | 7650 | | SNA CLA | |
| 0231 | 5271 | | JMP CRLF | /YES |
| 0232 | 5276 | | JMP TAB1A | /NONE OF THESE-TAB TO 1 |
| 0233 | 4061 | PGO, | JMS SPIGNR | /IGNORE LEADING SPACES |
| 0234 | 4320 | | JMS PUNCH | /PUNCH |
| 0235 | 4054 | | JMS GET | /GET NEXT |
| 0236 | 1024 | | TAD CHAR | /COMMA? |
| 0237 | 1053 | | TAD MCOMMA | |
| 0240 | 7640 | | SZA CLA | |
| 0241 | 5234 | | JMP PGO+1 | /NO |
| 0242 | 4320 | | JMS PUNCH | /YES PUNCH IT |

| | | | | |
|------|------|---------|------------|-----------------------------|
| | | /PAGE 3 | | |
| 0243 | 1023 | TAB1, | TAD FIELD1 | /IN FIELD? |
| 0244 | 7650 | | SNA CLA | |
| 0245 | 4352 | | JMS TABIT | /NO- TAB OVER |
| 0246 | 4061 | | JMS SPIGNR | /IGNORE SPACES |
| 0247 | 1024 | PG01, | TAD CHAR | |
| 0250 | 1022 | | TAD MSPACE | |
| 0251 | 7650 | | SNA CLA | /SPACE? |
| 0252 | 5266 | | JMP SPUNCH | /PUNCH IT AND IGNORE SPACES |
| 0253 | 1024 | | TAD CHAR | |
| 0254 | 1035 | | TAD MCR | /CR? |
| 0255 | 7650 | | SNA CLA | |
| 0256 | 5271 | | JMP CRLF | /YES |
| 0257 | 1024 | | TAD CHAR | |
| 0260 | 1034 | | TAD MSLASH | /"/"? |
| 0261 | 7650 | | SNA CLA | |
| 0262 | 5302 | | JMP TAB3 | /YES TAB TO STOP 3 |
| 0263 | 4320 | PG02, | JMS PUNCH | /PUNCH IT |
| 0264 | 4054 | | JMS GET | /GET NEXT |
| 0265 | 5247 | | JMP PG01 | |
| 0266 | 4320 | SPUNCH, | JMS PUNCH | |
| 0267 | 4061 | | JMS SPIGNR | |
| 0270 | 5247 | | JMP PG01 | |
| 0271 | 1036 | CRLF, | TAD CARRTN | /PUNCH CR, LF |
| 0272 | 4331 | | JMS PUN | |
| 0273 | 1037 | | TAD LFEED | |
| 0274 | 4331 | | JMS PUN | |
| 0275 | 5203 | | JMP START | |
| 0276 | 1023 | TAB1A, | TAD FIELD1 | /TAB TO STOP 1 |
| 0277 | 7650 | | SNA CLA | |
| 0300 | 4352 | | JMS TABIT | |
| 0301 | 5263 | | JMP PG02 | |
| 0302 | 1031 | TAB3, | TAD M3 | /IN FIELD 3? |
| 0303 | 1023 | | TAD FIELD1 | |
| 0304 | 7650 | | SNA CLA | |
| 0305 | 5311 | | JMP ECHO | /YES |
| 0306 | 4352 | | JMS TABIT | /NO TAB OVER 1 |
| 0307 | 5302 | | JMP TAB3 | |
| 0310 | 4054 | | JMS GET | |
| 0311 | 4320 | ECHO, | JMS PUNCH | /ECHO UNTIL CR |
| 0312 | 4054 | | JMS GET | |
| 0313 | 1024 | | TAD CHAR | |
| 0314 | 1035 | | TAD MCR | |
| 0315 | 7650 | | SNA CLA | |
| 0316 | 5271 | | JMP CRLF | |
| 0317 | 5311 | | JMP ECHO | |

| | | | | |
|------|------|---------|--------------|------------------------|
| 0320 | 0000 | /PAGE 4 | 0 | |
| 0321 | 1024 | PUNCH, | TAD CHAR | /PUNCH AND COUNT |
| 0322 | 4331 | | JMS PUN | |
| 0323 | 2025 | | ISZ FCOUNT | |
| 0324 | 5720 | | JMP I PUNCH | |
| 0325 | 1030 | | TAD MTAB1 | |
| 0326 | 3025 | | DCA FCOUNT | |
| 0327 | 2023 | | ISZ FIELD1 | |
| 0330 | 5720 | | JMP I PUNCH | |
| 0331 | 0000 | PUN, | 0 | /PUNCH CHARACTER IN AC |
| 0332 | 3351 | | DCA TEMP | |
| 0333 | 7604 | | CLA 0SR | |
| 0334 | 7700 | | SMA CLA | /OUTPUT?? |
| 0335 | 5344 | | JMP TGO | /33ASR |
| 0336 | 1351 | | TAD TEMP | /PUNCH IT H.SPEED |
| 0337 | 6026 | | PLS | |
| 0340 | 6021 | | PSF | |
| 0341 | 5340 | | JMP .-1 | |
| 0342 | 7200 | | CLA | |
| 0343 | 5731 | | JMP I PUN | |
| 0344 | 1351 | TGO, | TAD TEMP | |
| 0345 | 6046 | | TLS | |
| 0346 | 6041 | | TSF | |
| 0347 | 5346 | | JMP .-1 | |
| 0350 | 5342 | | JMP TGO-2 | |
| 0351 | 0000 | TEMP, | 0 | |
| 0352 | 0000 | TABIT, | 0 | |
| 0353 | 7604 | | CLA 0SR | /TABS OR SPACES |
| 0354 | 7010 | | RAR | |
| 0355 | 7620 | | SNL CLA | |
| 0356 | 5367 | | JMP SPACIT | /SPACES |
| 0357 | 1027 | | TAD TAB | |
| 0360 | 4331 | | JMS PUN | |
| 0361 | 1026 | | TAD C377 | |
| 0362 | 4331 | | JMS PUN | |
| 0363 | 1030 | | TAD MTAB1 | |
| 0364 | 3025 | | DCA FCOUNT | |
| 0365 | 2023 | | ISZ FIELD1 | |
| 0366 | 5752 | | JMP I TABIT | |
| 0367 | 1046 | SPACIT, | TAD SPACE | |
| 0370 | 4331 | | JMS PUN | |
| 0371 | 2025 | | ISZ FCOUNT | |
| 0372 | 5367 | | JMP SPACIT | |
| 0373 | 5363 | | JMP SPACIT-4 | |

/PAGE 5
/READ ROUTINES

*400

| | | | | |
|------|------|-------|------------|------------------------------|
| 0400 | 0000 | HIRD, | 0 | |
| 0401 | 7604 | | CLA 0SR | /INPUT SOURCE? |
| 0402 | 7700 | | SMA CLA | |
| 0403 | 5256 | | JMP K00 | /33ASR |
| 0404 | 2040 | | ISZ Z RCNT | /IS BUFFER EMPTY? |
| 0405 | 5247 | | JMP GCHR | /NO |
| 0406 | 2050 | | ISZ Z SWT1 | /DID TAPE END? |
| 0407 | 7410 | | SKP | |
| 0410 | 5252 | | JMP ENBF | /YES - GO HALT |
| 0411 | 3050 | | DCA Z SWT1 | |
| 0412 | 1020 | | TAD Z RBGN | /YES - READ IN A BUFFER FULL |
| 0413 | 3010 | | DCA Z 10 | |
| 0414 | 1042 | | TAD Z RKON | |
| 0415 | 3040 | | DCA Z RCNT | |
| 0416 | 6014 | RDIN, | RFC | /GET A CHARACTER |
| 0417 | 6047 | | DCA Z TEM6 | /START COUNTING |
| 0420 | 2047 | | ISZ Z TEM6 | /CCOUNTER OVERFLOW |
| 0421 | 7410 | | SKP | |
| 0422 | 5235 | | JMP FULL | /YES - END OF TAPE |
| 0423 | 6011 | | RSF | /NO - FLAG SET YET? |
| 0424 | 5220 | | JMP .-4 | /NO - COUNT AGAIN |
| 0425 | 7200 | | CLA | /YES - READ IN CHARACTER |
| 0426 | 6012 | | RRB | |
| 0427 | 4264 | | JMS TEST | /TEST INPUT |
| 0430 | 5216 | | JMP RDIN | /IGNORE IT |
| 0431 | 3410 | | DCA I 10 | /STORE IT |
| 0432 | 2040 | | ISZ Z RCNT | /IS BUFFER FULL YET? |
| 0433 | 5216 | | JMP RDIN | /NO |
| 0434 | 5241 | | JMP FULL | /YES |

/PAGE 6

| | | | | |
|------|------|-------|------------|-----------------------------|
| 0435 | 7240 | FUL1, | CLA CMA | |
| 0436 | 3050 | | DCA Z SWT1 | /TAPE RAN OFF END |
| 0437 | 1040 | | TAD Z RCNT | /SET END OF TAPE SWITCH |
| 0440 | 7041 | | CMA IAC | |
| 0441 | 1042 | FULL, | TAD Z RKON | |
| 0442 | 7450 | | SNA | |
| 0443 | 5252 | | JMP ENBF | |
| 0444 | 3040 | | DCA Z RCNT | |
| 0445 | 1020 | | TAD Z RBGN | /RESET COUNTER FOR READ |
| 0446 | 3012 | | DCA Z 12 | |
| 0447 | 1412 | GCHR, | TAD I Z 12 | /PICK UP A CHAR FROM BUFFER |
| 0450 | 3024 | | DCA CHAR | |
| 0451 | 5600 | | JMP I HIRD | /YES |
| 0452 | 7240 | ENBF, | CLA CMA | |
| 0453 | 3040 | | DCA Z RCNT | |
| 0454 | 7402 | | HLT | |
| 0455 | 5254 | | JMP .-1 | |
| 0456 | 6031 | KGO, | KSF | |
| 0457 | 5256 | | JMP .-1 | |
| 0460 | 6036 | | KRB | |
| 0461 | 4264 | | JMS TEST | /TEST INPUT |
| 0462 | 5256 | | JMP KGO | /IGNORE IT |
| 0463 | 5250 | | JMP GCHR+1 | |
| 0464 | 0000 | TEST, | 0 | /TEST INPUT |
| 0465 | 3024 | | DCA CHAR | /RETURN TO CALL+1 |
| 0466 | 1024 | | TAD CHAR | /IF TO BE IGNORED |
| 0467 | 1043 | | TAD MLFEED | /IS IT LINEFEED? |
| 0470 | 7650 | | SNA CLA | |
| 0471 | 5664 | | JMP I TEST | /YES-IGNORE |
| 0472 | 1024 | | TAD CHAR | |
| 0473 | 1044 | | TAD MRBOUT | /IS IT RUBOUT? |
| 0474 | 7650 | | SNA CLA | |
| 0475 | 5664 | | JMP I TEST | /YES-IGNORE |
| 0476 | 1024 | | TAD CHAR | |
| 0477 | 1045 | | TAD MTAB | /IS IT TAB? |
| 0500 | 7640 | | SZA CLA | |
| 0501 | 5304 | | JMP .+3 | /NO GET ORIGINAL |
| 0502 | 1046 | | TAD SPACE | /YES-CHANGE TO SPACE |
| 0503 | 7410 | | SKP | |
| 0504 | 1024 | | TAD CHAR | |
| 0505 | 2264 | | ISZ TEST | |
| 0506 | 5664 | | JMP I TEST | |

/PAGE 7

Digital-8-21-U-Sym

*600

Page 11

/ROUTINE TO READ A LINE


| | | | | |
|------|------|--------|--------------|------------------------------|
| 0600 | 0000 | READL, | 0 | |
| 0601 | 3052 | | DCA EQFLAG | |
| 0602 | 3051 | | DCA COMMFL | |
| 0603 | 3257 | | DCA SFLAG | /0 TO ALL FLAGS |
| 0604 | 1265 | | TAD BUF1 | |
| 0605 | 3013 | | DCA 13 | /SET BUFFER POINTER |
| 0606 | 4421 | RG0, | JMS I HIRDPT | /READ A CHARACTER |
| 0607 | 1024 | | TAD CHAR | |
| 0610 | 7450 | | SNA | |
| 0611 | 5260 | | JMP PG07 | /PUNCH BLANK |
| 0612 | 1032 | | TAD MFFFEED | /FORM FEED? |
| 0613 | 7450 | | SNA | |
| 0614 | 5260 | | JMP PG07 | /YES-PUNCH IT |
| 0615 | 1033 | | TAD MLEADR | /CODE 200? |
| 0616 | 7650 | | SNA CLA | |
| 0617 | 5260 | | JMP PG07 | /YES PUNCH IT |
| 0620 | 1024 | | TAD CHAR | |
| 0621 | 3413 | | DCA I 13 | /NONE::STORE IT |
| 0622 | 1024 | | TAD CHAR | |
| 0623 | 1034 | | TAD MSLASH | /IS IT SLASH? |
| 0624 | 7640 | | SZA CLA | |
| 0625 | 5230 | | JMP .+3 | /NO |
| 0626 | 2257 | | ISZ SFLAG | /YES SET COMMENT FLAG |
| 0627 | 5206 | | JMP RG0 | |
| 0630 | 1024 | | TAD CHAR | |
| 0631 | 1264 | | TAD MEQUL | /IS IT EQUAL SIGN? |
| 0632 | 7650 | | SNA CLA | |
| 0633 | 5247 | | JMP SET1 | /SET =FLAG IF NOT IN COMMENT |
| 0634 | 1024 | | TAD CHAR | |
| 0635 | 1053 | | TAD MCOMMA | /IS IT A COMMA? |
| 0636 | 7650 | | SNA CLA | |
| 0637 | 5253 | | JMP SET2 | /SET ,FLAG IF NOT IN COMMENT |
| 0640 | 1024 | | TAD CHAR | |
| 0641 | 1035 | | TAD MCR | /IS IT CR? |
| 0642 | 7640 | | SZA CLA | |
| 0643 | 5206 | | JMP RG0 | /NO:CONTINUE |
| 0644 | 1265 | | TAD BUF1 | /YES SET UP BUFFER |
| 0645 | 3013 | | DCA 13 | |
| 0646 | 5600 | | JMP I READL | /EXIT |
| 0647 | 1257 | SET1, | TAD SFLAG | |
| 0650 | 7650 | | SNA CLA | |
| 0651 | 2052 | | ISZ EQFLAG | |
| 0652 | 5206 | | JMP RG0 | |
| 0653 | 1257 | SET2, | TAD SFLAG | |
| 0654 | 7650 | | SNA CLA | |
| 0655 | 2051 | | ISZ COMMFL | |
| 0656 | 5206 | | JMP RG0 | |
| 0657 | 0000 | SFLAG, | 0 | |
| 0660 | 1024 | PG07, | TAD CHAR | |
| 0661 | 4663 | | JMS I .+2 | |
| 0662 | 5206 | | JMP RG0 | |
| 0663 | 0331 | | PUN | |
| 0664 | 7503 | MEQUL, | -275 | |
| 0665 | 0666 | BUF1, | +.1 | |

BUFI 0065
 CARRTN 0036
 CHAR 0024
 COMMFL 0051
 CRLF 0271
 C377 0026
 ECHO 0311
 ENDF 0452
 EQFLAG 0052
 FCOUNT 0025
 FIELD1 0023
 FULL 0441
 FUL1 0435
 GCHR 0447
 GET 0054
 HIRD 0400
 HIRDPT 0021
 KGO 0456
 LFEEED 0037
 MASTER 0041
 MCOMMA 0053
 MCR 0035
 MEQUL 0664
 MFFEEED 0032
 MLEADR 0033
 MLFEED 0043
 MRBOUT 0044
 MSLASH 0034
 MSPACE 0022
 MTAB 0045
 MTAB1 0030
 M3 0031
 PGO 0233
 PG01 0247
 PG02 0263
 PG07 0660
 PUN 0331
 PUNCH 0320
 RBGN 0020
 RCNT 0040
 RDIN 0416
 RDLINE 0070
 READL 0600
 RGO 0606
 RKON 0042
 SET1 0647
 SET2 0653
 SFLAG 0657
 SPACE 0046
 SPACIT 0367


SPIGNR 0061
 SPUNCH 0266
 START 0203
 SWT1 0050
 TAB 0027
 TABIT 0352
 TAB1 0243
 TAB1A 0276
 TAB3 0302
 TEMP 0351
 TEM6 0047
 TEST 0464
 TGO 0344

11.1 Flow Charts

11.1.1 Main Algorithm



11.1.2 Read-Line Subroutine



12. REFERENCES (Not Applicable)
13. ADDENDA (Not Applicable)